/*
 * Copyright 2012 Rivoli team
 *
 * Licensed under the Apache License, Version 2.0 (the "License");
 * you may not use this file except in compliance with the License.
 * You may obtain a copy of the License at
 *
 * http://www.apache.org/licenses/LICENSE-2.0
 *
 * Unless required by applicable law or agreed to in writing, software
 * distributed under the License is distributed on an "AS IS" BASIS,
 * WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
 * See the License for the specific language governing permissions and
 * limitations under the License.
 */
package com.googlecode.rivoli.integration

import com.googlecode.aluminumproject.configuration.DefaultConfiguration
import com.googlecode.aluminumproject.parsers.aluscript.AluScriptParser
import com.googlecode.aluminumproject.resources.AbstractTemplateFinder
import com.googlecode.rivoli.reader.aluminum.AluminumDefinitionReader

import org.testng.annotations.BeforeMethod

abstract class AluminumIntegrationTest extends IntegrationTest {
	private ByteArrayOutputStream out

	private static Map<String, String> templates

	@BeforeMethod
	void 'create templates'() {
		templates = [:]
	}

	static Closure getConfigurables() {
		{->
			'aluminum definition reader'(AluminumDefinitionReader, (AluminumDefinitionReader.ALUMINUM_PARAMETERS): [
				(DefaultConfiguration.TEMPLATE_FINDER_CLASS): MapTemplateFinder.name,
				(AluScriptParser.AUTOMATIC_NEWLINES): 'false'
			])
		}
	}

	static class MapTemplateFinder extends AbstractTemplateFinder {
		InputStream find(String name) {
			new ByteArrayInputStream(templates[name].bytes)
		}
	}

	protected void addTemplate(String name, String template) {
		templates[name] = template.stripIndent()
	}

	protected void setTemplate(String template) {
		addTemplate('test', template)
	}

	protected void generate(String templateName = 'test', Closure generationSettings) {
		super.generate({
			name = templateName
		} << generationSettings)
	}

	protected ByteArrayOutputStream createOut() {
		out = new ByteArrayOutputStream()
	}

	protected String getOutput() {
		out.toString()
	}
}